
[bookmark: _Toc83250293]
Intellectual Property Rights
Essential patents
[bookmark: IPR_3GPP]IPRs essential or potentially essential to normative deliverables may have been declared to ETSI. The information pertaining to these essential IPRs, if any, is publicly available for ETSI members and non-members, and can be found in ETSI SR 000 314: "Intellectual Property Rights (IPRs); Essential, or potentially Essential, IPRs notified to ETSI in respect of ETSI standards", which is available from the ETSI Secretariat. Latest updates are available on the ETSI Web server (https://ipr.etsi.org/).
Pursuant to the ETSI IPR Policy, no investigation, including IPR searches, has been carried out by ETSI. No guarantee can be given as to the existence of other IPRs not referenced in ETSI SR 000 314 (or the updates on the ETSI Web server) which are, or may be, or may become, essential to the present document.
Trademarks
The present document may include trademarks and/or tradenames which are asserted and/or registered by their owners. ETSI claims no ownership of these except for any which are indicated as being the property of ETSI, and conveys no right to use or reproduce any trademark and/or tradename. Mention of those trademarks in the present document does not constitute an endorsement by ETSI of products, services or organizations associated with those trademarks.
[bookmark: _Toc83250294]Foreword
This ETSI Standard (ES) has been produced by ETSI Technical Committee Methods for Testing and Specification (MTS).
The present document is part 8 of a multi-part deliverable. Full details of the entire series can be found in part 1 [1].	Comment by Philip Makedonski: Update Part 1 accordingly.
Update Parts 4 and 7 (extensions)	Comment by Finn Kristoffersen: Part 1, 4, and 7 updated with references to Part 8.
[bookmark: _Toc83250295]Modal verbs terminology
In the present document "shall", "shall not", "should", "should not", "may", "need not", "will", "will not", "can" and "cannot" are to be interpreted as described in clause 3.2 of the ETSI Drafting Rules (Verbal forms for the expression of provisions).
"must" and "must not" are NOT allowed in ETSI deliverables except when used in direct citation.
[bookmark: _Toc83250296]
1	Scope
The present document specifies the concrete textual syntax of the Test Description Language (TDL). The intended use of the present document is to serve as the basis for the development of textual TDL tools and TDL specifications. The meta-model of TDL and the meanings of the meta-classes are described in ETSI ES 203 119-1 [1].
NOTE:	OMG®, UML®, OCL™ and UTP™ are the trademarks of OMG (Object Management Group). This information is given for the convenience of users of the present document and does not constitute an endorsement by ETSI of the products named.	Comment by Philip Makedonski: Update
[bookmark: _Toc83250297]2	References
[bookmark: _Toc83250298]2.1	Normative references
References are either specific (identified by date of publication and/or edition number or version number) or non‑specific. For specific references, only the cited version applies. For non-specific references, the latest version of the referenced document (including any amendments) applies.
Referenced documents which are not found to be publicly available in the expected location might be found at https://docbox.etsi.org/Reference. 
NOTE:	While any hyperlinks included in this clause were valid at the time of publication, ETSI cannot guarantee their long term validity.
The following referenced documents are necessary for the application of the present document.
[bookmark: REF_ES203119_1][1]	ETSI ES 203 119-1 (V1.6.1): "Methods for Testing and Specification (MTS); The Test Description Language (TDL); Part 1: Abstract Syntax and Associated Semantics".
[bookmark: _Toc83250299]2.2	Informative references
References are either specific (identified by date of publication and/or edition number or version number) or non‑specific. For specific references, only the cited version applies. For non-specific references, the latest version of the referenced document (including any amendments) applies.
NOTE:	While any hyperlinks included in this clause were valid at the time of publication, ETSI cannot guarantee their long term validity.
The following referenced documents are not necessary for the application of the present document but they assist the user with regard to a particular subject area.
[bookmark: REF_ES202553][bookmark: REF_XTEXT_DOCUMENTATION][i.1]	Xtext?	Comment by Philip Makedonski: Reference to Xtext?	Comment by Martti Käärik: I guess https://www.eclipse.org/Xtext/documentation/301_grammarlanguage.html
[bookmark: REF_TS136523_1][i.2]	ETSI TS 136 523-1 (V10.2.0): "LTE; Evolved Universal Terrestrial Radio Access (E-UTRA) and Evolved Packet Core (EPC); User Equipment (UE) conformance specification; Part 1: Protocol conformance specification (3GPP TS 36.523-1 version 10.2.0 Release 10)".
[bookmark: REF_TS186011_2][i.3]	ETSI TS 186 011-2: "Core Network and Interoperability Testing (INT); IMS NNI Interoperability Test Specifications (3GPP Release 10); Part 2: Test descriptions for IMS NNI Interoperability".
[bookmark: _Toc83250300]3	Definition of terms, symbols and abbreviations
[bookmark: _Toc83250301]3.1	Terms
For the purposes of the present document, the terms given in ETSI ES 203 119-1 [1] and the following apply:
grammar: TODO
derivation: TODO
(production) rule: TODO
terminal: TODO
[bookmark: _Toc83250302]3.2	Symbols
Void?
[bookmark: _Toc83250303]3.3	Abbreviations
For the purposes of the present document, the following abbreviations apply:
BNF	Backus-Naur Form
EBNF	Extended Backus-Naur Form
IMS	IP Multimedia Subsystem
IUT	Implementation Under Test
OCL	Object Constraint Language
SUT	System Under Test
TDL	Test Description Language
[bookmark: _Toc83250304]4	Basic principles
[bookmark: _Toc83250305]4.1	Introduction
The meta-model of the Test Description Language is specified in ETSI ES 203 119-1 [1]. The presentation format of the meta-model can be different according to the needs of the users or the requests of the domain, where the TDL is applied. These presentation formats can either be text-oriented or graphic-oriented and may cover all the functionalities of the TDL meta-model or just a part of it, which is relevant to satisfy the needs of a specific application domain.
The present document specifies a concrete textual syntax that provides a textual representation for the whole functionality of the TDL meta-model.	Comment by Philip Makedonski: Confirm this is the case. -> No, certain parts are syntactically excluded -> update text
The document specifies the TDL textual file format, where the textual representations of the instances of the TDL meta-classes may be placed. A textual representation may contain keywords, delimiters, and textual labels within a defined structure. The rules, how these strucutres shall be interpreted, are described by means of Extended Backus-Naur Form (EBNF)-like expressions. In particular, in addition to the syntactical structure, the EBNF-like expressions also indicate how the textual labels and structures are mapped to the TDL meta-model.
[bookmark: _Toc83250306]4.2	Document Structure
The present document specifies the concrete textual syntax of the Test Description Language (TDL).
Clause 5 specifies the TDL textual file format.	Comment by Philip Makedonski: Will this be necessary?
Clause 6 specifies the concrete production rules defined for the TDL meta-classes. (The meta-model of TDL and the meanings of the meta-classes are described in ETSI ES 203 119-1 [1].)
Foundation (clause 6.1)
Data (clause 6.2)
Time (clause 6.3)
Test Configuration (clause 6.4)
Test Behaviour (clause 6.5)
At the end of the document several examples illustrating the features of the TDL Textual Syntax can be found.
[bookmark: _Toc83250307]4.3	Grammar Language	Comment by Philip Makedonski: TODO -> Martti
[bookmark: _Toc49244492][bookmark: _Toc83250308]4.3.1	Overview
The rules that define the textual syntax of the TDL are described in present document using the grammar language of the Xtext framework. In addition to defining the lexical structure of the TDL syntax the grammar language also provides means for mapping those textual constructs to the TDL meta-model. Additional rules such as identity resolution and linking are described where applicable to provide complete mapping of textual TDL to the TDL model.
The grammar of textual TDL is composed of a number of grammar rules organized in a tree. The grammar structure follows the logical structure of the TDL meta-model and the root of the grammar is the 'Package' production rule. Production rules are used to construct model objects and assign values to the properties of those objects. Production rules consist of keywords (character literals) and calls to production rules, data type rules and terminal rules (which correspond to tokens of text).
The following clauses describe the syntax of the grammar language. See Xtext documentation for further details [i.1].
[bookmark: _Toc83250309]4.3.2	Operators
Various operators are used in grammar rule definitions to specify the order and cardinality of keywords and rule calls. Terminal rule specific operators are used to express various textual constructs. Production rule specific operators are used to define assignments and cross-references.
Following operators are used in all rule definitions:
'?' indicates that preceding construct shall occur 0 or 1 times;
'*' indicates that preceding construct shall occur 0 or more times;
'+' indicates that preceding construct shall occur 1 or more times;
'|' is used between alternative constructs and
'(' and ')' is are used to group constructs defined in between.
[bookmark: _Hlk83218617]Following operators are used in terminal rule definitions:
'!' is used to negate a construct;
'->' is used to indicate that everything is ignored until the following construct is detected;
'..' is used between charaters to define a range and
'.' denotes any character.
Following operators are used in production rule definitions:
'=' is used to define a simple assignment of a right hand construct to a property on the left;
'+=' is used for assigning (adding to) multi-valued property;
'?=' is used for assigning the value 'true' to a Boolean property on the condition that the right hand side construct is present and
'[', '|' and ']' is are used to define a cross-reference.
Various special symbols are included in the grammar definitions of production rules that are included solely as implementation detail (to help the generation of a parser for textual TDL) and do not alter the definition of the syntax. Such symbols include '->' and '=>'.
[bookmark: _Toc83250310]4.3.3	Terminal rules and keywords
Lexical tokens in the TDL grammar are either keywords of character sequences that are matched and consumed by terminal rules during parsing. In the grammar definition, keywords are placed between apostrophes (').
Terminal rule declarations start with the keyword 'terminal' followed by the rule name (in upper-case letters by convention). The rule name is followed by 'returns' keyword and the reference to a data type that is used for creating a value using the consumed token.
The definition of the rule starts with a colon (':') and ends with a semi-colon (';'). Terminal rule definitions consist of terminal rule calls (indicated by rule name), characters and operators.
EXAMPLE:	terminal INT returns EInt: ('0'..'9')+;
Some terminal rules (such as comments and whitespace) are defined as hidden in TDL grammar and corresponding text shall be allowed anywhere in textual TDL (outside of tokens).
TODO: put all terminal rules here or in clause 5?	Comment by Martti Käärik: We should include the imported terminals somewhere.
	Comment by Martti Käärik: In 5.5
[bookmark: _Toc83250311]4.3.4	Production rules
Production rules are used to create model objects or data values. The rules that return a data type instead of a meta-class are known as data type rules.
Production rule declarations start with the rule name followed by 'returns' keyword and the reference to the meta-class that defines the object that is produced by the rule. The definition of the rule starts with a colon (':') and ends with a semi-colon (';'). Production rule definitions consist of rule calls, keywords, and operators.
EXAMPLE:	Comment returns tdl::Comment:
    'Note:' body=EString
;
An assignment is defined as a property name followed by an assignment operator (see clause 4.3.2) followed by a rule call (name of production or data type rule) or a cross-reference. A cross-reference is defined as a meta-class reference followed by '|' and a terminal rule call that defines the format for the identifier. The cross-reference definition is placed between square brackets ('[' and ']').
EXAMPLE:	Annotation returns tdl::Annotation:
    '@' key=[tdl::AnnotationType|Identifier]
    (':' value=EString)?
;
Production rule calls may also be used without assignment. In that case the model object that is returned from the calling rule is the one that is created in the called rule.
Production rules may be created as fragments by prefixing the declaration with the 'fragment' keyword. In that case the rule does not produce an object by itself but rather assigns to properties of the object that is created in the calling rule. Fragment rules are always unassigned.
[bookmark: _Toc49244495][bookmark: _Toc83250312]4.3.6	Examples	Comment by Martti Käärik: Examples are within the text above. Could remove this.
[bookmark: _Toc83250313]4.4	Conformance
For an implementation claiming to conform to this version of the TDL Concrete Textual Syntax, all features specified in the present document and in ETSI ES 203 119-1 [1] shall be implemented consistently with the requirements given in the present document and ETSI ES 203 119-1 [1].
[bookmark: _Toc83250314]5	General rules
[bookmark: _Toc83250315]5.1	Identities and references	Comment by Philip Makedonski: Make sure that description matches what is supported with regard to naming and qualified names
In TDL models, references between objects are based on unique identifiers that are generated by the modelling framework and stored in model files. Such identifiers are generally hidden from the user. In textual TDL, all attributes shall be part of the text document and the use of such identifiers is not feasible.
In textual TDL, objects are identified by 'name' or 'qualifiedName' property. The allowed values for the 'name' property are restricted by the terminal rule 'ID' (see clause 5.5). The exception to this rule is made for objects that are predefined in TDL and are mapped to special symbols in textual TDL (such as AnyValue).
If the 'name' property shall have a value that is equal to a keyword in textual TDL then that value shall be prefixed with '^' in the text.
[bookmark: _Toc83250316]5.2	Models and importing
TDL objects stored in a single file are collectively referred to as model. Both the TDL model and textual TDL allow single 'Package' object as the root of the model. Thus, logically the root package of a TDL file is a TDL model.	Comment by Martti Käärik: “TDL model” is used as the term for EMF/XMI based serialization here and elsewhere. However it might be difficult to understand it in this context.
Naming of textual TDL files and the location of those files is out of the scope of the present document. Implementations of the textual TDL shall provide means to make TDL models available for importing.
Imported 'Package's shall be referred to by the value of 'qualifiedName' property.
[bookmark: _Toc83250317]5.3	Linking
Linking refers to the phase in the compilation process of textual TDL where name-based cross-references are resolved to actual objects that they represent. By default, linking utilizes object identities are described in clause 5.1.
In some cases where explicit cross-references are not required by the grammar rules, the linking may apply context specific logic to assign references to object properties. Such cases are described in relevant clauses.
[bookmark: _Toc83250318]5.4	Alternative syntaxes
Although the keywords are specified with certain case (lower-case or title-case) in the present document, the case itself is not prescribed. Therefore, an implementation can be case-insensitive as well. It is recommended that users apply a consistent case nonetheless.
The delimiters for 'Block's and other constructs are specified in an abstract manner with the 'BEGIN' and 'END' terminal symbols. While the default assumption is that these terminal symbols are mapped to left and right braces ('{' and '}'), referred to as 'brace-based' syntax, an alternative implementation using white space indentation is also possible, where synthetic delimiters for the beginning and end of indented parts shall be used instead, referred to as 'indentation-based' syntax. Besides the replacement of the 'BEGIN' and 'END' no other differences shall be present between implementations of the 'brace-based' and 'indentation-based' syntax. Left and right braces ('{' and '}') shall be used in certain contexts even within the 'indentation-based' syntax, e.g. for 'TimeConstraint's and data-related 'Constraint's.
The examples in this document conform to the default assumption. Additional examples illustrating the indentation-based syntax are included in Annex B.x. 
[bookmark: _Toc83250319]5.5	Terminals
TODO
[bookmark: _Toc83250320]5.6	File format	Comment by Philip Makedonski: Probably removed or moved to TR.
Anything on encoding, tile extensions, other aspects? -> No. File naming is outside the scope of this document.
TODO
[bookmark: _Toc83250321]6	Production Rules
[bookmark: _Toc83250322]6.1	Foundation
[bookmark: _Toc83250323]6.1.1	Element
Concrete Textual Notation	Comment by Philip Makedonski: Predefined fragments, data rules, terminals
Beginning of clause 6/5

Fragments shall be placed in the corresponding element (abstract)
fragment AnnotationFragment returns tdl::Element:
    (annotation+=Annotation)*
;

fragment AnnotationCommentFragment returns tdl::Element:
    (comment+=Comment)*
    (annotation+=Annotation)*
;

fragment NameFragment returns tdl::Element:
    'Name:' name=Identifier
;

//Use: inline block - within other blocks, e.g. with other contents 
fragment WithCommentFragment returns tdl::Element:
    'with'
    BEGIN
        (comment+=Comment)+
        //TODO: Shall this be '*'?
    END
;

fragment WithNameFragment returns tdl::Element:
    'with'
    BEGIN
        NameFragment
    END
;

Comments
This is an abstract metaclass, therefore no textual representation is defined for the element. The concrete textual notation represents reusable fragments that can be embedded in the concrete textual notation of metaclasses inheriting from this metaclass. 
The different fragments are used in different contexts.
Examples	Comment by Philip Makedonski: Add brace and indentation?
Add example identifiers
	Note: "Example test objective"
	@Example

	with {
		Note: "Comment on nested package"
	}

	with {
		Name: anOptionalNameForElementWithoutMandatoryName
	}

[bookmark: _Toc83250324]6.1.2	NamedElement
Concrete Textual Notation
Void.
Comments
This is an abstract metaclass, therefore no textual representation is defined for the element. 
Examples
Void.
[bookmark: _Toc83250325]6.1.3	ElementImport
Concrete Textual Notation
ElementImport returns tdl::ElementImport:
    AnnotationCommentFragment
    'Import'
    ('all' | 
        (importedElement+=[tdl::PackageableElement|Identifier] 
            ( "," importedElement+=[tdl::PackageableElement|Identifier])*
        )
    )
    'from' importedPackage=[tdl::Package|QIdentifier]
;

Comments
No comments.
Examples
	Import all from NestedPackage
	Import NestedAnnotation from NestedPackage

[bookmark: _Toc83250326]6.1.4	Package
Concrete Textual Notation	Comment by Philip Makedonski: What about semi-colons or other delimiters?
Use only if necessary or use everywhere? -> Decide once more complete picture emerges. 
Quite certainly they will have to remain at least in some place, the more specific question is whether they will be optional in other places where they are not mandatory? (can be a problem of its own)	Comment by Philip Makedonski: No need for semi colons, commas needed in many places still
Package returns tdl::Package:
    AnnotationFragment
    'Package' name=Identifier
    BEGIN
        (^import+=ElementImport)*
        (packagedElement+=PackageableElement)*
        (nestedPackage+=Package)*
    END
    WithCommentFragment?
;

Comments	Comment by Philip Makedonski: Put constraints and other information here.
'Annotation's applied to the 'Package' shall be defined within the 'Package' or imported in the 'Package' from other 'Package's even as the 'Annotation's are applied on the “outside” of the 'Package'.
Examples	Comment by Philip Makedonski: Both indentation and brace based? -> stick with one, provide comprehensive examples for both delimiters in the annex.	Comment by Philip Makedonski: Focus on braces at first.
@NestedAnnotation
Package Foundation {
	Note: "Example imports from nested (or other) package"
	Import all from NestedPackage
	Import NestedAnnotation from NestedPackage
	
	Note: "Annotate examples"
	AnnotationType Example
	
	Note: "Annotate standardised constructs"
	AnnotationType Standard 

	Package NestedPackage {
		AnnotationType NestedAnnotation
	} 
	with {
		Note: "Comment on nested package"
	}

} with {
	Note: "Comment on foundation"
} 
[bookmark: _Toc83250327]6.1.5	PackageableElement
Concrete Textual Notation
PackageableElement returns tdl::PackageableElement:
    (AnnotationType | TestObjective
     | ConstraintType
     | DataResourceMapping | DataElementMapping
     | SimpleDataType_Impl | SimpleDataInstance 
     | StructuredDataType | StructuredDataInstance 
     | CollectionDataType | CollectionDataInstance
     | ProcedureSignature 
     | Action_Impl | Function 
     | PredefinedFunction 
     | EnumDataType
     | Time 
     | ComponentType | GateType 
     | TestConfiguration 
     | TestDescription
    )
;

Comments
This is an abstract metaclass, the textual representation depends on the concrete types indicated as alternative derivations.
Examples
Void.
[bookmark: _Toc83250328]6.1.6	Comment
Concrete Textual Notation
Comment returns tdl::Comment:
    'Note:' body=EString
;

Comments
In different contexts, depending on the fragments being used, a 'Comment' may be defined before the 'Element' or within a 'with' block.
'Comment's are syntactically excluded from certain constructs, e.g. 'DataUse'.
Examples
	Note: "Example test objective"
	TestObjective TO_Foundation 

[bookmark: _Toc83250329]6.1.7	Annotation
Concrete Textual Notation
Annotation returns tdl::Annotation:
    '@' key=[tdl::AnnotationType|Identifier]
    (':' value=EString)?
;

Comments
'Annotation's are syntactically excluded from certain constructs, e.g. 'DataUse'.
Examples
	@Example
	TestObjective TO_Foundation 

[bookmark: _Toc83250330]6.1.8	AnnotationType
Concrete Textual Notation
AnnotationType returns tdl::AnnotationType:
    AnnotationCommentFragment
    'AnnotationType' name=Identifier
    extension=Extension?
;

Comments
No comments.
Examples
	@Example
	@Version: "0.1"
	TestObjective TO_Foundation 

[bookmark: _Toc83250331]6.1.9	TestObjective
Concrete Textual Notation
TestObjective returns tdl::TestObjective:
    AnnotationCommentFragment
    'TestObjective' name=Identifier
    (BEGIN
        ('Description:' description=EString)?
        ('References:' objectiveURI+=EString (',' objectiveURI+=EString)*)?
    END)?
;

Comments
No comments.
Examples
	TestObjective TO_Foundation {
		Description: "Illustrate the definition of a test objectives"
		References: "This package.",
			"A base document", 
			"Another source"
	}

[bookmark: _Toc83250332]6.1.10	Extension
Concrete Textual Notation
Extension returns tdl::Extension:
    //NOTE: No annotations or comments
    'extends' extending=[tdl::PackageableElement|Identifier]
;

Comments
'Annotation's and 'Comment's are syntactically excluded.
Examples
	Type Post (
		String title,
		String date
	) 

	Type TaggedPost extends Post (
		Tags tags
	)

[bookmark: _Toc83250333]6.1.11	ConstraintType
Concrete Textual Notation
ConstraintType returns tdl::ConstraintType: 
    AnnotationCommentFragment
    'Constraint' name=Identifier
;

Comments
No comments.
Examples
	@Example
	Constraint HexString
	@Example
	Constraint DateString
	@Standard
	Constraint Length

[bookmark: _Toc83250334]6.1.12	Constraint
Concrete Textual Notation
Constraint returns tdl::Constraint: 
    //NOTE: No annotations or comments
    type=[tdl::ConstraintType|Identifier]
    (':' quantifier+=LiteralValueUse (',' quantifier+=LiteralValueUse)+ )?
;

Comments
'Annotation's and 'Comment's are syntactically excluded.
Examples
	{Length : 1, 10} 
	Collection Posts of Post

[bookmark: _Toc83250335]6.2	Data	Comment by Philip Makedonski: Separate into Definition and Use?
[bookmark: _Toc83250336]6.2.1	DataResourceMapping
Concrete Textual Notation
DataResourceMapping returns tdl::DataResourceMapping:
    AnnotationCommentFragment
    'Use' resourceURI=EString
    'as' name=Identifier
;

Comments
No comments.
Examples
	Note: "Use external resource for the mapping"
	Use "API.yaml" as API

[bookmark: _Toc83250337]6.2.2	DataElementMapping
Concrete Textual Notation
DataElementMapping returns tdl::DataElementMapping:
    AnnotationCommentFragment
    'Map' mappableDataElement=[tdl::MappableDataElement|Identifier]
    ('to' elementURI=EString)?
    'in' dataResourceMapping=[tdl::DataResourceMapping|Identifier]
    'as' name=Identifier
    (BEGIN 
        //TODO: separator? 
        (parameterMapping+=ParameterMapping)*
    END)?
;

Comments
No comments.
Examples
	Note: "Map data elements to concrete data in external resource"
	Map Post to "api/post" in API as PostMapping {
		title -> "post::title"
		date -> "post::date"
	}

[bookmark: _Toc83250338]6.2.3	ParameterMapping
Concrete Textual Notation
ParameterMapping returns tdl::ParameterMapping:
    AnnotationCommentFragment
    parameter=[tdl::Parameter|Identifier]
    '->' parameterURI=EString
;

Comments
No comments.
Examples
	title -> "post::title"
	date -> "post::date"

[bookmark: _Toc83250339]6.2.3a	DataType
Concrete Textual Notation
fragment ConstraintFragment returns tdl::DataType:
    //TODO: put braces into constraint?
    LBrace (constraint+=Constraint)+ RBrace
;

Comments
This is an abstract metaclass, therefore no textual representation is defined for the element. The concrete textual notation represents reusable fragments that can be embedded in the concrete textual notation of metaclasses inheriting from this metaclass. 
Examples
Void.

[bookmark: _Toc83250340]6.2.4	SimpleDataType
Concrete Textual Notation
SimpleDataType_Impl returns tdl::SimpleDataType:
    AnnotationCommentFragment
    ConstraintFragment?
    'Type' name=Identifier
;

Comments
No comments.
Examples
	@Standard
	Type String
	@Standard
	Type Integer
	@Standard
	Type Verdict

[bookmark: _Toc83250341]6.2.5	SimpleDataInstance
Concrete Textual Notation
SimpleDataInstance returns tdl::SimpleDataInstance:
    AnnotationCommentFragment
    dataType=[tdl::DataType|Identifier]
    name=Identifier
;

Comments
No comments.
Examples
	@Standard
	Verdict fail
	String authToken
	Integer sessionId

[bookmark: _Toc83250342]6.2.6	StructuredDataType
Concrete Textual Notation
StructuredDataType returns tdl::StructuredDataType:
    AnnotationCommentFragment
    ConstraintFragment?
    'Type' name=Identifier
    extension=Extension?
    '(' (member+=Member (',' member+=Member)*)? ')'
;

Comments
No comments.
Examples
	Type Post (
		String title,
		String date
	) 

[bookmark: _Toc83250343]6.2.7	Member
Concrete Textual Notation
Member returns tdl::Member:
    AnnotationCommentFragment
    MemberConstraintFragment?
    (isOptional?='optional')?
    dataType=[tdl::DataType|Identifier]
    name=Identifier
;

fragment MemberConstraintFragment returns tdl::Member:
    //TODO: put braces into constraint?
    LBrace (constraint+=Constraint)+ RBrace
;

Comments
No comments.
Examples
		String title,
		Note: "Constraint for members"
		{DateString} String date,
		Note: "Optional members"
		optional Tags tags

[bookmark: _Toc83250344]6.2.8	StructuredDataInstance
Concrete Textual Notation
StructuredDataInstance returns tdl::StructuredDataInstance:
    AnnotationCommentFragment
    dataType=[tdl::DataType|Identifier]
    name=Identifier
    //TODO: use keywords for unassigned member treatment?
    (unassignedMember=UnassignedMemberTreatment)?
    '(' (memberAssignment+=MemberAssignment (',' memberAssignment+=MemberAssignment)*)? ')'
;

Comments
No comments.
Examples
	Post firstReport ? (
		title = "first report",
		date = "today"
	)

[bookmark: _Toc83250345]6.2.9	MemberAssignment
Concrete Textual Notation
MemberAssignment returns tdl::MemberAssignment:
    AnnotationCommentFragment
    member=[tdl::Member|Identifier]
    '=' memberSpec=StaticDataUse
;

Comments
No comments.
Examples
	title = "first report",
	date = firstReport.date,
	tags = ?

[bookmark: _Toc83250346]6.2.10	CollectionDataType
Concrete Textual Notation
CollectionDataType returns tdl::CollectionDataType:
    AnnotationCommentFragment
    ConstraintFragment?
    'Collection' name=Identifier
    'of' itemType=[tdl::DataType|Identifier]
;

Comments
No comments.
Examples
	Collection Posts of Post
	Collection Tags of Tag

[bookmark: _Toc83250347]6.2.11	CollectionDataInstance
Concrete Textual Notation
CollectionDataInstance returns tdl::CollectionDataInstance:
    AnnotationCommentFragment
    dataType=[tdl::DataType|Identifier] 
    name=Identifier
    //TODO: separators?
    '[' item+=StaticDataUse* ']'
;

Comments
No comments.
Examples
	Tags usefulReportsFilter [useful report]
	Posts allPosts [ 
		a Post(title="first post", date="yesterday") 
		a Post(title="second post", date="today")
		firstReport
		secondReport
	]	

[bookmark: _Toc83250348]6.2.12	ProcedureSignature
Concrete Textual Notation
ProcedureSignature returns tdl::ProcedureSignature:
    AnnotationCommentFragment
    'Signature' name=Identifier
    '(' parameter+=ProcedureParameter ( "," parameter+=ProcedureParameter)* ')'
;

Comments
No comments.
Examples
    Signature publish (in Post post, out Integer postId)

[bookmark: _Toc83250349]6.2.13	ProcedureParameter
Concrete Textual Notation
ProcedureParameter returns tdl::ProcedureParameter:
    AnnotationFragment
    kind=ParameterKind
    dataType=[tdl::DataType|Identifier]
    name=Identifier
    WithCommentFragment?
;

Comments
No comments.
Examples
    in Post post
    out Integer postId

[bookmark: _Toc83250350]6.2.14	ParameterKind
Concrete Textual Notation
enum ParameterKind returns tdl::ParameterKind:
    In = 'in' | Out = 'out' | Exception = 'exception';

Comments
No comments.
Examples
Void.
[bookmark: _Toc83250351]6.2.15	Parameter
Concrete Textual Notation
Void.
Comments
This is an abstract metaclass, therefore no textual representation is defined for the element. 
Examples
Void.
[bookmark: _Toc83250352]6.2.16	FormalParameter
Concrete Textual Notation
FormalParameter returns tdl::FormalParameter:
    AnnotationFragment
    dataType=[tdl::DataType|Identifier]
    name=Identifier
    WithCommentFragment?
;

Comments
No comments.
Examples
    @Encrypted
    Post post
    @Unique
    Integer postId

[bookmark: _Toc83250353]6.2.17	Variable
Concrete Textual Notation
Variable returns tdl::Variable:
    AnnotationCommentFragment
    'variable' dataType=[tdl::DataType|Identifier]
    name=Identifier
    WithCommentFragment?
;

Comments
No comments.
Examples
	variable Binary authToken 

[bookmark: _Toc83250354]6.2.18	Action
Concrete Textual Notation
Action_Impl returns tdl::Action:
    AnnotationCommentFragment
    'Action' name=Identifier
    ('(' formalParameter+=FormalParameter ( "," formalParameter+=FormalParameter)* ')')?
    (':' body=EString)?
;

Comments
No comments.
Examples
    Action reset
    Action clean: "Cleaning procedure: Wash hands, wear mask and gloves, open windows."
    Action reload(Posts posts): "Reloading procedure: Clear all posts, reset, reload posts."

[bookmark: _Toc83250355]6.2.19	Function
Concrete Textual Notation
Function returns tdl::Function:
    AnnotationCommentFragment
    'Function' name=Identifier
    ('(' formalParameter+=FormalParameter ( "," formalParameter+=FormalParameter)* ')')?
    'returns' returnType=[tdl::DataType|Identifier]
    (':' body=EString)?
;

Comments
No comments.
Examples
    Function categoriseReport(Post post, Tags tags) returns Post: "Categorise with text mining"

[bookmark: _Toc83250356]6.2.20	UnassignedMemberTreatment
Concrete Textual Notation
enum UnassignedMemberTreatment returns tdl::UnassignedMemberTreatment:
    AnyValue = '?' | AnyValueOrOmit = '*';

Comments
No comments.
Examples
	Post firstReport ? (
		title = "first report",
		date = "today"
	)

[bookmark: _Toc83250357]6.2.21	PredefinedFunction
Concrete Textual Notation
PredefinedFunction returns tdl::PredefinedFunction:
    AnnotationCommentFragment
    'Predefined' 
    (name=PredefinedIdentifierBinary
        | name=PredefinedIdentifierNot
        | name=PredefinedIdentifierSize
    )
    ('returns' returnType=[tdl::DataType|Identifier])?
;

PredefinedIdentifierBinary returns ecore::EString:
    '+' | '-' | '*' | '/' | 'mod' 
  | '>' | '<' | '>=' | '<=' //TODO: fix extra '>=' also in TDLan2
  | '==' | '!=' | 'and' | 'or' | 'xor'
;

PredefinedIdentifierNot returns ecore::EString:
    'not'
;

PredefinedIdentifierSize returns ecore::EString:
    'size'
;

Comments
The 'PredefinedFunction's shall be provided as a standard library.
Examples
	Predefined ==
	Predefined !=
	Predefined +

[bookmark: _Toc83250358]6.2.22	EnumDataType
Concrete Textual Notation
EnumDataType returns tdl::EnumDataType: 
    AnnotationCommentFragment
    'Enumerated' name=Identifier
    BEGIN
        //TODO: without type specification? separator
        (value+=SimpleDataInstance)+
    END
;

Comments
No comments.
Examples
	Enumerated Tag {
		Tag useful
		Tag interesting
		Tag report
	}

[bookmark: _Toc83250359]6.2.23	DataUse
Concrete Textual Notation
DataUse returns tdl::DataUse:
    //NOTE: No name, annotations or comments
    StaticDataUse | =>DynamicDataUse
;

fragment ReductionFragment returns tdl::DataUse:
    (->reduction+=CollectionReference)?
    ('.' reduction+=MemberReference)*
;

fragment ParameterBindingFragment returns tdl::DataUse:
    =>'(' BareParameterBindingFragment ')'
;

fragment BareParameterBindingFragment returns tdl::DataUse:
    =>(argument+=ParameterBinding ( "," argument+=ParameterBinding)*)
;

Comments
This is an abstract metaclass, the textual representation depends on the concrete types indicated as alternative derivations. The reusable fragments can be embedded in the concrete textual notation of metaclasses inheriting from this metaclass.
 'Annotation's and 'Comment's, as well as the 'name' property, are syntactically excluded from all concrete types.
Examples
Void.
[bookmark: _Toc83250360]6.2.24	ParameterBinding
Concrete Textual Notation
ParameterBinding returns tdl::ParameterBinding:
    parameter=[tdl::Parameter|Identifier]
    '=' dataUse=DataUse
;

Comments
No comments.
Examples
Void.
[bookmark: _Toc83250361]6.2.25	MemberReference
Concrete Textual Notation
MemberReference returns tdl::MemberReference:
    member=[tdl::Member|Identifier] 
    (->'[' collectionIndex=DataUse ']')? 
;

CollectionReference returns tdl::MemberReference:
    '[' collectionIndex=DataUse ']'
;

Comments
The 'CollectionReference' derivation is applicable in case only a collection reference is needed, for example, immediately after a 'DataUse' with a type resolving to a 'CollectionDataType'.
Examples
TODO

[bookmark: _Toc83250362]6.2.26	StaticDataUse
Concrete Textual Notation
StaticDataUse returns tdl::StaticDataUse:
    DataInstanceUse
    | AnonymousDataInstanceUse 
    | SpecialValueUse
    | LiteralValueUse
;

Comments
This is an abstract metaclass, the textual representation depends on the concrete types indicated as alternative derivations.
Examples
Void.
[bookmark: _Toc83250363]6.1.3	DataInstanceUse
Concrete Textual Notation
DataInstanceUse returns tdl::DataInstanceUse:
    (dataInstance=[tdl::DataInstance|Identifier]
    |
    ('a' dataType=[tdl::StructuredDataType|Identifier]))
    UnassignedFragment?
    ParameterBindingFragment?
    ReductionFragment
;

AnonymousDataInstanceUse returns tdl::DataInstanceUse:
    UnassignedFragment?
    ParameterBindingFragment
;

fragment UnassignedFragment returns tdl::DataInstanceUse:
    //TODO: added angle brackets for disambiguation
    =>'<' unassignedMember=UnassignedMemberTreatment '>'
;

Comments
The 'AnonymousDataInstanceUse' derivation is applicable in case no 'DataInstance' and no 'DataType' are specified.
Examples
TODO

[bookmark: _Toc83250364]6.1.3	SpecialValueUse
Concrete Textual Notation
SpecialValueUse returns tdl::SpecialValueUse:
    OmitValue | AnyValue | AnyValueOrOmit 
;

Comments
This is an abstract metaclass, the textual representation depends on the concrete types indicated as alternative derivations.
Examples
Void.
[bookmark: _Toc83250365]6.1.3	AnyValue
Concrete Textual Notation
AnyValue returns tdl::AnyValue:
    name='?'
    ('of' 'type' dataType=[tdl::DataType|Identifier])?
;

Comments
No comments.
Examples
Void.
[bookmark: _Toc83250366]6.1.3	AnyValueOrOmit
Concrete Textual Notation
AnyValueOrOmit returns tdl::AnyValueOrOmit:
    name='*'
;

Comments
No comments.
Examples
Void.
[bookmark: _Toc83250367]6.1.3	OmitValue
Concrete Textual Notation
OmitValue returns tdl::OmitValue:
    name='omit'
;

Comments
No comments.
Examples
Void.
[bookmark: _Toc83250368]6.1.3	LiteralValueUse
Concrete Textual Notation
LiteralValueUse returns tdl::LiteralValueUse:
    (value=STRING | intValue=BIGINTEGER | boolValue=BOOLEAN)
    //TODO: add option to have the type after wards without any keyword e.g. 100 s?
    ('of' 'type' dataType=[tdl::DataType|Identifier])?
;

Comments
No comments.
Examples
TODO

[bookmark: _Toc83250369]6.1.3	DynamicDataUse
Concrete Textual Notation
DynamicDataUse returns tdl::DynamicDataUse:
    FunctionCall 
    | FormalParameterUse
    | VariableUse 
    | PredefinedFunctionCall
    | TimeLabelUse 
;

Comments
This is an abstract metaclass, the textual representation depends on the concrete types indicated as alternative derivations.
Examples
Void.
[bookmark: _Toc83250370]6.1.3	FunctionCall
Concrete Textual Notation
FunctionCall returns tdl::FunctionCall:
    //TODO: refine syntax
    'instance' 'returned' 'from' function=[tdl::Function|Identifier] 
    '(' BareParameterBindingFragment? ')'
    ReductionFragment
;

Comments
No comments.
Examples
TODO

[bookmark: _Toc83250371]6.1.3	FormalParameterUse
Concrete Textual Notation
FormalParameterUse returns tdl::FormalParameterUse:
    'parameter' parameter=[tdl::FormalParameter|Identifier]
    ParameterBindingFragment?
    ReductionFragment
;

Comments
No comments.
Examples
TODO

[bookmark: _Toc83250372]6.1.3	VariableUse
Concrete Textual Notation
VariableUse returns tdl::VariableUse:
    //TODO: try to adjust syntax so that -> is not necessary
    componentInstance=[tdl::ComponentInstance|Identifier]
    '->' variable=[tdl::Variable|Identifier]
    ParameterBindingFragment?
    ReductionFragment
;

Comments
No comments.
Examples
TODO

[bookmark: _Toc83250373]6.1.3	PredefinedFunctionCall
Concrete Textual Notation
PredefinedFunctionCall returns tdl::PredefinedFunctionCall:
    PredefinedFunctionCallSize
    | PredefinedFunctionCallNot 
    | PredefinedFunctionCallBinary
;

PredefinedFunctionCallSize returns tdl::PredefinedFunctionCall:
    function=[tdl::PredefinedFunction|PredefinedIdentifierSize] 
    '(' actualParameters+=DataUse ')'
;


PredefinedFunctionCallNot returns tdl::PredefinedFunctionCall:
    function=[tdl::PredefinedFunction|PredefinedIdentifierNot] 
    '(' actualParameters+=DataUse ')'
;

PredefinedFunctionCallBinary returns tdl::PredefinedFunctionCall:
    '('
    actualParameters+=DataUse 
    function=[tdl::PredefinedFunction|PredefinedIdentifierBinary] 
    actualParameters+=DataUse
    ')'
;

Comments
No comments.
Examples
   assert (size(allPosts)==4)
	assert not(client->authenticated)

[bookmark: _Toc83250374]6.3	Time
[bookmark: _Toc83250375]6.3.1	Time
Concrete Textual Notation
Time returns tdl::Time:
    AnnotationCommentFragment
    'Time' name=Identifier
;

Comments
No comments.
Examples
	Time seconds
	Time milliseconds

[bookmark: _Toc83250376]6.3.2	TimeLabel
Concrete Textual Notation
TimeLabel returns tdl::TimeLabel:
    //NOTE: No annotations or comments
    name=Identifier
;

Comments
'Annotation's and 'Comment's are syntactically excluded.
Examples
    publicationTime=now

[bookmark: _Toc83250377]6.3.3	TimeLabelUse
Concrete Textual Notation
TimeLabelUse returns tdl::TimeLabelUse:
    //NOTE: No reduction and parameter binding (constraint)
    //TODO: refine syntax
    'time' 'label' timeLabel=[tdl::TimeLabel|Identifier]
    ('.' kind=TimeLabelUseKind)?
;

Comments
Assignment of the 'reduction' and 'argument' properties is syntactically excluded.
Examples
    time label publicationTime
    time label publicationTime.last

[bookmark: _Toc83250378]6.3.4	TimeLabelUseKind
Concrete Textual Notation
enum TimeLabelUseKind returns tdl::TimeLabelUseKind:
    Last = 'last' | Previous = 'previous' | First = 'first'
;

Comments
No comments.
Examples
Void.

[bookmark: _Toc83250379]6.3.5	TimeConstraint
Concrete Textual Notation
TimeConstraint returns tdl::TimeConstraint:
    //NOTE: No annotations or comments
    timeConstraintExpression=DataUse
;

Comments
'Annotation's and 'Comment's are syntactically excluded.
Examples
Void.
[bookmark: _Toc83250380]6.3.6	Timer
Concrete Textual Notation
Timer returns tdl::Timer:
    AnnotationCommentFragment
    'timer' name=Identifier
;

Comments
No comments.
Examples
	timer global

[bookmark: _Toc83250381]6.3.7	TimeOperation
Concrete Textual Notation
TimeOperation returns tdl::TimeOperation: 
    Wait | Quiescence
;

Comments
This is an abstract metaclass, the textual representation depends on the concrete types indicated as alternative derivations.
Examples
Void.
[bookmark: _Toc83250382]6.3.8	Wait
Concrete Textual Notation
Wait returns tdl::Wait: 
    AtomicPrefixFragment
    'wait' 'for' period=DataUse
    ( 'on' componentInstance=[tdl::ComponentInstance|Identifier])
;

Comments
No comments.
Examples
    wait for 10 
    wait for 10 on client 

[bookmark: _Toc83250383]6.3.9	Quiescence
Concrete Textual Notation
Quiescence returns tdl::Quiescence: 
    AtomicPrefixFragment
    'quiet' 'for' period=DataUse 
    (  'on' (
            componentInstance=[tdl::ComponentInstance|Identifier] 
            | gateReference=[tdl::GateReference|IdentifierDot]
        )
    )?
;

Comments
No comments.
Examples
    quiet for 5
    quiet for 5 on server
    quiet for 5 on server.http

[bookmark: _Toc83250384]6.3.10	TimerOperation
Concrete Textual Notation
TimerOperation returns tdl::TimerOperation: 
    TimerStart | TimerStop | TimeOut
;

Comments
This is an abstract metaclass, the textual representation depends on the concrete types indicated as alternative derivations.
Examples
Void.
[bookmark: _Toc83250385]6.3.11	TimerStart
Concrete Textual Notation
TimerStart returns tdl::TimerStart:
    AtomicPrefixFragment
    'start' componentInstance=[tdl::ComponentInstance|Identifier]
    //TODO: use '->' instead?
    '.' timer=[tdl::Timer|Identifier]
    'for' period=DataUse
;

Comments
No comments.
Examples
    start client.global for 10

[bookmark: _Toc83250386]6.3.12	TimerStop
Concrete Textual Notation
TimerStop returns tdl::TimerStop:
    AtomicPrefixFragment
    'stop' componentInstance=[tdl::ComponentInstance|Identifier]
    '.' timer=[tdl::Timer|Identifier]
;

Comments
No comments.
Examples
    stop client.global

[bookmark: _Toc83250387]6.3.13	TimeOut
Concrete Textual Notation
TimeOut returns tdl::TimeOut:
    AtomicPrefixFragment
    'time' 'out' 'on' componentInstance=[tdl::ComponentInstance|Identifier]
    '.' timer=[tdl::Timer|Identifier]
;

Comments
No comments.
Examples
    time out on client.global

[bookmark: _Toc83250388]6.4	Test Configuration
[bookmark: _Toc83250389]6.4.1	GateType
Concrete Textual Notation
GateType returns tdl::GateType:
    AnnotationCommentFragment
    kind=GateTypeKind
    'Gate' name=Identifier
    (extension=Extension)?
    'accepts' dataType+=[tdl::DataType|Identifier] ( "," dataType+=[tdl::DataType|Identifier])*
;

Comments
No comments.
Examples
	Message Gate HTTP accepts Post, Posts
	
	Message Gate HTTPS extends HTTP accepts Binary
	
	Procedure Gate RPC accepts publish

[bookmark: _Toc83250390]6.4.2	GateTypeKind
Concrete Textual Notation
enum GateTypeKind returns tdl::GateTypeKind:
    Message = 'Message' | Procedure = 'Procedure';
Comments
No comments.
Examples
Void.
[bookmark: _Toc83250391]6.4.3	GateInstance
Concrete Textual Notation
GateInstance returns tdl::GateInstance:
    AnnotationCommentFragment
    'gate' type=[tdl::GateType|Identifier]
    name=Identifier
    WithCommentFragment?  
;

Comments
No comments.
Examples
		gate HTTP http
		gate RPC rpc

[bookmark: _Toc83250392]6.1.3	ComponentType
Concrete Textual Notation
ComponentType returns tdl::ComponentType:
    AnnotationCommentFragment
    'Component' name=Identifier
    (extension=Extension)?
    BEGIN
        (timer+=Timer)*
        (variable+=Variable)*
        (gateInstance+=GateInstance)* 
    END
;

Comments
No comments.
Examples
	Component Node {
		timer global
		variable Binary authToken
		variable Integer lastPostId 
		gate HTTP http
		gate RPC rpc
	}
	
	Component SecureNode extends Node {
		gate HTTPS https
	}

[bookmark: _Toc83250393]6.1.3	ComponentInstance
Concrete Textual Notation
ComponentInstance returns tdl::ComponentInstance:
    AnnotationCommentFragment
    'create' role=ComponentInstanceRole
    name=Identifier
    'of' 'type' type=[tdl::ComponentType|Identifier]
;

Comments
No comments.
Examples
		create SUT server of type Node 
		create Tester client of type Node

[bookmark: _Toc83250394]6.1.3	ComponentInstanceRole
Concrete Textual Notation
enum ComponentInstanceRole returns tdl::ComponentInstanceRole:
    SUT = 'SUT' | Tester = 'Tester';

Comments
No comments.
Examples
Void.
[bookmark: _Toc83250395]6.1.3	GateReference
Concrete Textual Notation
GateReference returns tdl::GateReference:
    //NOTE: No annotations or comments
    (name=Identifier '=')?
    component=[tdl::ComponentInstance|Identifier]
    '.' 
    gate=[tdl::GateInstance|Identifier]
;

Comments
No comments.
Examples
Void.
[bookmark: _Toc83250396]6.1.3	Connection
Concrete Textual Notation
Connection returns tdl::Connection:
    AnnotationCommentFragment
    'connect' endPoint+=GateReference 
    'to' endPoint+=GateReference
    WithNameFragment?
;

Comments
No comments.
Examples
		connect client.http to server.http
		connect cRPC=client.rpc to sRPC=server.rpc  

[bookmark: _Toc83250397]6.1.3	TestConfiguration
Concrete Textual Notation
TestConfiguration returns tdl::TestConfiguration:
    AnnotationCommentFragment
    'Configuration' name=Identifier
    BEGIN
        (componentInstance+=ComponentInstance)* 
        (connection+=Connection)* 
    END
;

Comments
No comments.
Examples
	Configuration base {
		create SUT server of type Node 
		create Tester client of type Node
		connect client.http to server.http
		connect cRPC=client.rpc to sRPC=server.rpc  
	}


[bookmark: _Toc83250398]6.5	Test Behaviour
[bookmark: _Toc83250399]6.5.1	TestDescription
Concrete Textual Notation
TestDescription returns tdl::TestDescription:
    TDPrefixFragment
    //TODO: consider better way to specify / indicate local ordering
    ('TestDescription' | isLocallyOrdered?='Test') 
    name=Identifier
    ('(' formalParameter+=FormalParameter ( "," formalParameter+=FormalParameter)* ')' )?
    'uses' testConfiguration=[tdl::TestConfiguration|Identifier]
    (behaviourDescription=BehaviourDescription)?
;

fragment TDPrefixFragment returns tdl::TestDescription:
    (comment+=Comment)*
    TDObjectiveFragment?
    (annotation+=Annotation)*
;

fragment TDObjectiveFragment returns tdl::TestDescription:
    'Objective:' testObjective+=[tdl::TestObjective|Identifier] 
     ( "," testObjective+=[tdl::TestObjective|Identifier])*
;

Comments
No comments.
Examples
	 @Example
    TestDescription publishNewRreport(Post cleanPost, Binary authRequest) uses base

	 @Example
    Test publishNewRreport(Post cleanPost, Binary authRequest) uses base

    Objective: CheckAuthToken
 	 TestDescription publishNewRreport(Post cleanPost, Binary authRequest)
    uses base { 
    		perform action : "Call administrator" on client
    }

[bookmark: _Toc83250400]6.5.2	BehaviourDescription
Concrete Textual Notation
BehaviourDescription returns tdl::BehaviourDescription:
    //NOTE: No annotations or comments
    behaviour=Behaviour
;

Comments
'Annotation's and 'Comment's, as well as the 'name' property, are syntactically excluded from 'BehaviourDescription'.
Examples
Void.
[bookmark: _Toc83250401]6.5.3	Behaviour
Concrete Textual Notation
Behaviour returns tdl::Behaviour:
    CombinedBehaviour | AtomicBehaviour
    //NOTE: Excluded via constraint:
    //PeriodicBehaviour
    //DefaultBehaviour | InterruptBehaviour | 
;

fragment WithBehaviourFragment returns tdl::Behaviour:
    'with'
    BEGIN
        NameFragment?
        ObjectiveFragment?
        (comment+=Comment)*
    END
;

fragment ObjectiveFragment returns tdl::Behaviour:
    'Objective:' testObjective+=[tdl::TestObjective|Identifier] ( "," testObjective+=[tdl::TestObjective|Identifier])*
;

Comments
The reusable fragments can be embedded in the concrete textual notation of metaclasses inheriting from this metaclass.
Examples
Void.
[bookmark: _Toc83250402]6.1.3	Block
Concrete Textual Notation
Block returns tdl::Block:
    //NOTE: No annotations or comments
    ('[' guard+=LocalExpression ( "," guard+=LocalExpression)* ']' )?
    BEGIN
        behaviour+=Behaviour+ 
    END
;

Comments
'Annotation's and 'Comment's, as well as the 'name' property, are syntactically excluded from 'Blocks'. 'Annotation's and 'Comment's can be assigned to the containing 'CombinedBehaviour's
Examples
Void.
[bookmark: _Toc83250403]6.1.3	LocalExpression
Concrete Textual Notation
LocalExpression returns tdl::LocalExpression:
    //NOTE: No annotations or comments
    expression=DataUse
    ('on' scope=[tdl::ComponentInstance|Identifier])?
;

LocalLoopExpression returns tdl::LocalExpression:
    //NOTE: No annotations or comments
    expression=DataUse 'times'
    ('on' scope=[tdl::ComponentInstance|Identifier])?
;

Comments
The 'LocalLoopExpression' derivation is only used within 'BoundedLoopBehaviour's. 'Annotation's and 'Comment's, as well as the 'name' property, are syntactically excluded from 'LocalExpression's.
Examples
Void.

[bookmark: _Toc83250404]6.1.3	CombinedBehaviour
Concrete Textual Notation
CombinedBehaviour returns tdl::CombinedBehaviour:
    (SingleCombinedBehaviour | MultipleCombinedBehaviour)
    =>WithCombinedFragment?
;

fragment WithCombinedFragment returns tdl::CombinedBehaviour:
    'with'
    BEGIN
        NameFragment?
        ObjectiveFragment?
        (comment+=Comment)*
        //TODO: periodic / exceptional as separate fragment(s) for combined behaviour?
        (periodic+=PeriodicBehaviour)*
        (exceptional+=ExceptionalBehaviour)*
    END
;

Comments
This is an abstract metaclass, the textual representation depends on the concrete types indicated as alternative derivations. The 'WithCombinedFragment' is always assigned to the innermost 'CombinedBehaviour'.
Examples
Void.
[bookmark: _Toc83250405]6.1.3	SingleCombinedBehaviour
Concrete Textual Notation
SingleCombinedBehaviour returns tdl::SingleCombinedBehaviour:
    CompoundBehaviour
    | BoundedLoopBehaviour
    | UnboundedLoopBehaviour 
    | OptionalBehaviour
;

Comments
This is an abstract metaclass, the textual representation depends on the concrete types indicated as alternative derivations.
Examples
Void.
[bookmark: _Toc83250406]6.1.3	CompoundBehaviour
Concrete Textual Notation
CompoundBehaviour returns tdl::CompoundBehaviour:
    AnnotationFragment
    block=Block
;

Comments
No comments.
Examples
        @Example
        [ "some expression" ] {
            perform action: "reload"
        }

        @Example
        {
            perform action: "reload"
        }

[bookmark: _Toc83250407]6.1.3	BoundedLoopBehaviour
Concrete Textual Notation
BoundedLoopBehaviour returns tdl::BoundedLoopBehaviour:
    AnnotationFragment
    'repeat' numIteration+=LocalLoopExpression (',' numIteration+=LocalLoopExpression)*
    block=Block
;

Comments
No comments.
Examples
        repeat [ "some expression" ] {
            perform action: "reload"
        }

        repeat [ "some expression" on client, "other expression" on server] {
            perform action: "reload"
        }

[bookmark: _Toc83250408]6.1.3	UnboundedLoopBehaviour
Concrete Textual Notation
UnboundedLoopBehaviour returns tdl::UnboundedLoopBehaviour:
    AnnotationFragment
    'repeat' block=Block
;

Comments
No comments.
Examples
        repeat 5 times {
            perform action: "reload"
        }

        repeat 5 times on client, 3 times on server {
            perform action: "reload"
        }


[bookmark: _Toc83250409]6.1.3	OptionalBehaviour
Concrete Textual Notation
OptionalBehaviour returns tdl::OptionalBehaviour:
    AnnotationFragment
    'optionally' block=Block
;

Comments
No comments.
Examples
        optionally {
            perform action: "reload"
        }


[bookmark: _Toc83250410]6.1.3	MultipleCombinedBehaviour
Concrete Textual Notation
MultipleCombinedBehaviour returns tdl::MultipleCombinedBehaviour:
    ConditionalBehaviour 
    | AlternativeBehaviour
    | ParallelBehaviour
;

Comments
This is an abstract metaclass, the textual representation depends on the concrete types indicated as alternative derivations.
Examples
Void.
[bookmark: _Toc83250411]6.1.3	ConditionalBehaviour
Concrete Textual Notation
ConditionalBehaviour returns tdl::ConditionalBehaviour:
    AnnotationFragment
    'if' block+=Block
    (=>('else' block+=Block)
    | (('else' 'if' block+=Block)* 
       ('else' block+=Block)))? 
;

Comments
The 'Block's are identifed to by the preceding keywords, where the first 'Block'  is referred to as an 'if' 'Block' and the following 'Block's are referred to as 'else' or 'else if' 'Blocks', respectively. An 'else' 'Block' shall always be attached to the innermost 'if' 'Block'.
Examples
        if [ "some expression" ] {
            perform action: "reload"
        }

        if [ "some expression" on client, "other expression" on server] {
            perform action: "reload"
        } else if [ "some expression" on client, "other expression" on server] {
            perform action: "backup"
        } else {
            perform action: "query"
        }


[bookmark: _Toc83250412]6.1.3	AlternativeBehaviour
Concrete Textual Notation
AlternativeBehaviour returns tdl::AlternativeBehaviour:
    AnnotationFragment
    'alternatively' block+=Block
    //TODO: enforce 2+? 
    ('or' block+=Block)* 
;

Comments
No comments.
Examples
      alternatively {
		 	server.http sends "error" to client.http
			perform action: "reload"
      } or {
         time out on client.global
         perform action: "reload conditionally"
      }

      alternatively [ "some expression"] {
		 	server.http sends "error" to client.http
			perform action: "reload"
      } or [ "some other expression"] {
         time out on client.global
         perform action: "reload conditionally"
      }

      alternatively [ "some expression" on client, "other expression" on server] {
		 	server.http sends "error" to client.http
			perform action: "reload"
      } or {
         time out on client.global
         perform action: "reload conditionally"
      }


[bookmark: _Toc83250413]6.1.3	ParallelBehaviour
Concrete Textual Notation
ParallelBehaviour returns tdl::ParallelBehaviour:
    AnnotationFragment
    'run' block+=Block 
    //TODO: enforce 2+?
    //TODO: simplify keywords / syntax
    ('in' 'parallel' 'to' block+=Block)?
    ('and' block+=Block)*
;

Comments
No comments.
Examples
      run {
			perform action: "reload"
		} in parallel to {
			perform action: "backup"
      } and {
			perform action: "query"
      }

      run [ "some expression" on client, "other expression" on server] {
			perform action: "reload"
		} in parallel to {
			perform action: "backup"
      } and [ "some expression" on client, "other expression" on server] {
			perform action: "query"
      }


[bookmark: _Toc83250414]6.1.3	ExceptionalBehaviour
Concrete Textual Notation
ExceptionalBehaviour returns tdl::ExceptionalBehaviour:
    DefaultBehaviour | InterruptBehaviour;

Comments
This is an abstract metaclass, the textual representation depends on the concrete types indicated as alternative derivations. 
Examples
Void.
[bookmark: _Toc83250415]6.1.3	DefaultBehaviour
Concrete Textual Notation
DefaultBehaviour returns tdl::DefaultBehaviour:
    AnnotationFragment
    'default' 
    ('on' guardedComponent=[tdl::ComponentInstance|Identifier])?
    block=Block
    WithBehaviourFragment?
;

Comments
No comments.
Examples
        repeat [ "some expression" on client, "other expression" on server] {
            perform action: "query" 
        } with {
        		Note: "Applies to combined behaviour"
            default {
					server.http sends "error" to client.http
					perform action: "reload"
            } with {
            	Note: "Applies to default"
            }
            default on client {
            	time out on client.global
					perform action: "reload"
            }
        }

[bookmark: _Toc83250416]6.1.3	InterruptBehaviour
Concrete Textual Notation
InterruptBehaviour returns tdl::InterruptBehaviour:
    AnnotationFragment
    'interrupt'
    ('on' guardedComponent=[tdl::ComponentInstance|Identifier])?
    block=Block
    WithBehaviourFragment?
;

Comments
No comments.
Examples
        repeat [ "some expression" on client, "other expression" on server] {
            perform action: "query" 
        } with {
            interrupt on client [ "some condition" ] {
                time out on client.global
                perform action: "reload conditionally"
            }
        }


[bookmark: _Toc83250417]6.1.3	PeriodicBehaviour
Concrete Textual Notation
PeriodicBehaviour returns tdl::PeriodicBehaviour:
    AnnotationFragment
    'every'
    (period+=LocalExpression (',' period+=LocalExpression)*)
    block=Block
    WithBehaviourFragment?
;

Comments
No comments.
Examples
        repeat [ "some expression" on client, "other expression" on server] {
            perform action: "query"
        } with {
            every 5 [ "some expression" on client, "other expression" on server] { 
					perform action: "reload"
            }
        }


[bookmark: _Toc83250418]6.1.3	AtomicBehaviour
Concrete Textual Notation
AtomicBehaviour returns tdl::AtomicBehaviour:
    (TimerOperation
    | TimeOperation
    | Break | Stop
    | VerdictAssignment | Assertion
    | Interaction
    | TestDescriptionReference 
    | ActionBehaviour 
    | Assignment)
    WithAtomicFragment? 
;

fragment AtomicPrefixFragment returns tdl::AtomicBehaviour:
    ObjectiveFragment?
    (comment+=Comment)*
    (annotation+=Annotation)*
;

fragment WithAtomicFragment returns tdl::AtomicBehaviour:
    'with'
    BEGIN
        NameFragment?
        TimeLabelFragment?
        TimeConstraintFragment?
    END
;

fragment TimeLabelFragment returns tdl::AtomicBehaviour:
    timeLabel=TimeLabel '=now' 
;

fragment TimeConstraintFragment returns tdl::AtomicBehaviour:
    LBrace timeConstraint+=TimeConstraint ( "," timeConstraint+=TimeConstraint)* RBrace
;

Comments
The reusable fragments can be embedded in the concrete textual notation of metaclasses inheriting from this metaclass.

Examples
    		Objective: CheckAuthToken

        	with {
        		publicationTime=now
        	}

        	with {
        		{(time label publicationTime == 2)}
        	}

[bookmark: _Toc83250419]6.1.3	Break
Concrete Textual Notation
Break returns tdl::Break:
    {tdl::Break}
    AtomicPrefixFragment
    'break'
;

Comments
No comments.
Examples
    break


[bookmark: _Toc83250420]6.1.3	Stop
Concrete Textual Notation
Stop returns tdl::Stop:
    {tdl::Stop}
    AtomicPrefixFragment
    'terminate'
;

Comments
No comments.
Examples
    terminate


[bookmark: _Toc83250421]6.1.3	VerdictAssignment
Concrete Textual Notation
VerdictAssignment returns tdl::VerdictAssignment:
    AtomicPrefixFragment
    'set' 'verdict' 'to' verdict=DataUse
;

Comments
No comments.
Examples
    set verdict to fail


[bookmark: _Toc83250422]6.1.3	Assertion
Concrete Textual Notation
Assertion returns tdl::Assertion:
    {tdl::Assertion}
    AtomicPrefixFragment
    'assert' condition=DataUse
    ('on' componentInstance=[tdl::ComponentInstance|Identifier])?
    //TODO: simplify syntax
    ('otherwise' 'set' 'verdict' 'to' otherwise=DataUse)?
;

Comments
No comments.
Examples
    assert (client->authToken==referenceToken)
    assert (client->authToken==referenceToken) on client
    assert (client->authToken=="101010") otherwise set verdict to fail

[bookmark: _Toc83250423]6.1.3	Interaction
Concrete Textual Notation
Interaction returns tdl::Interaction: 
    Message | ReceiveMessage | ProcedureCall | ProcedureCallResponse
;

Comments
This is an abstract metaclass, the textual representation depends on the concrete types indicated as alternative derivations.
Examples
Void.
[bookmark: _Toc83250424]6.1.3	Message
Concrete Textual Notation
Message returns tdl::Message:
    AtomicPrefixFragment
    sourceGate=[tdl::GateReference|IdentifierDot]
    ('sends' | (isTrigger?='triggers')) 
    argument=DataUse
    'to' target+=TargetMessage ( "," target+=TargetMessage)* 
;

ReceiveMessage returns tdl::Message:
    AtomicPrefixFragment
    //NOTE: Only single target possible
    target+=ReceiveTargetMessage //( "," target+=ReceiveTargetMessage)*
    'receives' (isTrigger?='trigger')? 
    argument=DataUse
    'from' sourceGate=[tdl::GateReference|IdentifierDot]
;

Comments
A 'ReceiveMessage' can be used as an alternative way to specify a 'Message' by switching the source and target gates. Only a single 'Target' can be specified when within a 'ReceiveMessage'. If multiple 'Target's need to be specified, the 'Message' derivation shall be used instead..
Examples
    Note: "Single target"
    server.http sends a Binary to client.http

    Note: "Multiple targets"
    server.http sends a Binary to client.http, bridge.http

    Note: "Single target assignment"
    server.http sends a Binary to  
    	client.http assigned to authToken

    Note: "Multiple targets with assignment"
    server.http sends a Binary to  
    	client.http assigned to authToken,
		bridge.http assigned to authToken 

    Note: "Alternative notation without assignment"
    client.http receives a Binary from server.http 

    Note: "Alternative notation with assignment"
    authToken = client.http receives a Binary from server.http 

[bookmark: _Toc83250425]6.1.3	Target
Concrete Textual Notation
TargetMessage returns tdl::Target:
    //NOTE: No name, annotations or comments
    targetGate=[tdl::GateReference|IdentifierDot]
    (valueAssignment+=ValueAssignmentMessage)?
;

ReceiveTargetMessage returns tdl::Target:
    //NOTE: No name, annotations or comments
    (valueAssignment+=ReceiveValueAssignmentMessage)?
    targetGate=[tdl::GateReference|IdentifierDot]
;

TargetProcedure returns tdl::Target:
    //NOTE: No name, annotations or comments
    targetGate=[tdl::GateReference|IdentifierDot]
    //TODO: check what the correct syntax shall be
    (valueAssignment+=ValueAssignmentProcedure ("," valueAssignment+=ValueAssignmentProcedure)*)?
;

Comments
The alternative derivations are used in the respective contexts.
'Annotation's and 'Comment's, as well as the 'name' property, are syntactically excluded from all concrete types.
Examples
Void.

[bookmark: _Toc83250426]6.1.3	ValueAssignment
Concrete Textual Notation
ValueAssignmentMessage returns tdl::ValueAssignment:
    //NOTE: No name, annotations or comments
    //TODO: check what the correct syntax shall be
    //'where' 'it' 'is' 
    'assigned' 'to' variable=[tdl::Variable|Identifier]
;

ReceiveValueAssignmentMessage returns tdl::ValueAssignment:
    //NOTE: No name, annotations or comments
    variable=[tdl::Variable|Identifier] '='
;

ValueAssignmentProcedure returns tdl::ValueAssignment:
    //NOTE: No name, annotations or comments
    //TODO: check what the correct syntax shall be
    'where' parameter=[tdl::Parameter|Identifier] 
    'is' 'assigned' 'to' variable=[tdl::Variable|Identifier]
;

Comments
The alternative derivations are used in the respective contexts.
'Annotation's and 'Comment's, as well as the 'name' property, are syntactically excluded from all concrete types.
Examples
    assigned to authToken
    where postId is assigned to lastPostId 


[bookmark: _Toc83250427]6.1.3	ProcedureCall
Concrete Textual Notation
ProcedureCall returns tdl::ProcedureCall:
    AtomicPrefixFragment
    sourceGate=[tdl::GateReference|IdentifierDot]
    'calls' signature=[tdl::ProcedureSignature|Identifier]
    '('argument+=ParameterBinding ("," argument+=ParameterBinding)*')'
    'on' target+=TargetProcedure
;

ProcedureCallResponse returns tdl::ProcedureCall:
    AtomicPrefixFragment
    //TODO: this shall probably be switched
    (replyTo=[tdl::ProcedureCall|Identifier] ':')?
    sourceGate=[tdl::GateReference|IdentifierDot]
    'responds' 'with' signature=[tdl::ProcedureSignature|Identifier]
    '('argument+=ParameterBinding ("," argument+=ParameterBinding)*')'
    'to' target+=TargetProcedure
;

Comments
A 'ProcedureCallResponse' shall be used to specify a response after a 'ProcedureCall'. The 'ProcedureCall' shall be specified with an assigned 'name'-property within the 'WithCombinedFragment'.
Examples
    client.rpc calls publish(post=firstReport) on server.rpc with {
    	Name: publishCall 
    }
    publishCall: server.rpc responds with publish(postId=1) to client.rpc 
    publishCall: server.rpc responds with publish(postId=1) to client.rpc 
    	where postId is assigned to lastPostId 

[bookmark: _Toc83250428]6.1.3	TestDescriptionReference
Concrete Textual Notation
TestDescriptionReference returns tdl::TestDescriptionReference:
    AtomicPrefixFragment
    'execute' testDescription=[tdl::TestDescription|Identifier]
    ('(' actualParameter+=ParameterBinding ( "," actualParameter+=ParameterBinding)* ')' )?
    (BEGIN
        componentInstanceBinding+=ComponentInstanceBinding 
        (',' componentInstanceBinding+=ComponentInstanceBinding)+
    END)?
;

Comments
No comments.
Examples
    execute publishAll
    execute publishClean(
    	cleanPost = a Post<?>(title = "Cleaner post"), 
    	authRequest = "00111001"
    )
    execute publishClean(
    	cleanPost = a Post<?>(title = "Cleanest post"), 
    	authRequest = parameter authRequest
    ) {
    	bind client to client,
    	bind server to server
    } 

[bookmark: _Toc83250429]6.1.3	ComponentInstanceBinding
Concrete Textual Notation
ComponentInstanceBinding returns tdl::ComponentInstanceBinding:
    AnnotationCommentFragment
    'bind' formalComponent=[tdl::ComponentInstance|Identifier]
    'to' actualComponent=[tdl::ComponentInstance|Identifier]
;

Comments
No comments.
Examples
    bind client to webClient,
    bind server to webServer


[bookmark: _Toc83250430]6.1.3	ActionBehaviour
Concrete Textual Notation
ActionBehaviour returns tdl::ActionBehaviour: 
    ActionReference | InlineAction
;

Comments
This is an abstract metaclass, the textual representation depends on the concrete types indicated as alternative derivations.
Examples
Void.
[bookmark: _Toc83250431]6.1.3	ActionReference
Concrete Textual Notation
ActionReference returns tdl::ActionReference:
    AtomicPrefixFragment
    'perform' action=[tdl::Action|Identifier]
    ('(' actualParameter+=ParameterBinding ( "," actualParameter+=ParameterBinding)* ')' )?
    ('on' componentInstance=[tdl::ComponentInstance|Identifier])?
;

Comments
No comments.
Examples
    perform reset on server
    perform reload(posts = allPosts) on server

[bookmark: _Toc83250432]6.1.3	InlineAction
Concrete Textual Notation
InlineAction returns tdl::InlineAction:
    AtomicPrefixFragment
    'perform' 'action' ':' body=EString
    ('on' componentInstance=[tdl::ComponentInstance|Identifier])?
;

Comments
No comments.
Examples
    perform action : "Call administrator" on client

[bookmark: _Toc83250433]6.1.3	Assignment
Concrete Textual Notation
Assignment returns tdl::Assignment:
    AtomicPrefixFragment
    variable=VariableUse
    '=' expression=DataUse
    ('on' componentInstance=[tdl::ComponentInstance|Identifier])?
;

Comments
No comments.
Examples
	client->authToken = "101010"
	client->authToken = instance returned from generateToken(seed = 12) on client



[bookmark: _Toc83250434]Annex A (normative):
Complete Textual Syntax Grammar	Comment by Philip Makedonski: Include in overview.	Comment by Philip Makedonski: Also electronic? Or both?

[bookmark: _Toc83250435]A.0	Overview
Flatten all sub-clauses?
[bookmark: _Toc83250436]A.1	Conventions	Comment by Philip Makedonski: Should be covered in Clause 4. Remove?
The notations is based on the Extended Backus-Naur Form (EBNF) notation. The EBNF representation may be used either as a concrete syntax reference for TDL for end users or as input to a parser generator tool. Table A.1 defines the syntactic conventions that are to be applied when reading the EBNF rules.	Comment by Philip Makedonski: Update / remove
Table A.1: Syntax definition conventions used
	::=
	is defined to be

	abc
	the non-terminal symbol abc

	abc xyz
	abc followed by xyz

	abc | xyz
	alternative (abc or xyz)

	[abc]
	0 or 1 instance of abc

	{abc}+
	1 or more instances of abc

	{abc}
	0 or more instances of abc

	'a'-'z'
	all characters from a to z

	(...)
	denotes a textual grouping

	'abc'
	the terminal symbol abc

	;
	production terminator

	\
	the escape character



[bookmark: _Toc83250437]A.2	Production Rules	Comment by Philip Makedonski: Condensed version of all parts of clause 6 -> Electronic reference?

Also avoid using numbers, they do not really add any benefits but rather cause considerable maintenance overhead.
//# Foundation

//Element : Abstract

//NamedElement : Abstract

//Needs to be the first rule
Package returns tdl::Package:
    AnnotationFragment
    'Package' name=Identifier
    BEGIN
        (^import+=ElementImport)*
        (packagedElement+=PackageableElement)*
        (nestedPackage+=Package)*
    END
    WithCommentFragment?
;

PackageableElement returns tdl::PackageableElement:
    (AnnotationType | TestObjective
     | ConstraintType
     | DataResourceMapping | DataElementMapping
     | SimpleDataType_Impl | SimpleDataInstance 
     | StructuredDataType | StructuredDataInstance 
     | CollectionDataType | CollectionDataInstance
     | ProcedureSignature 
     | Action_Impl | Function 
     | PredefinedFunction 
     | EnumDataType
     | Time 
     | ComponentType | GateType 
     | TestConfiguration 
     | TestDescription
    )
;

ElementImport returns tdl::ElementImport:
    AnnotationCommentFragment
    'Import'
    ('all' | 
        (importedElement+=[tdl::PackageableElement|Identifier] 
            ( "," importedElement+=[tdl::PackageableElement|Identifier])*
        )
    )
    'from' importedPackage=[tdl::Package|QIdentifier]
;

Comment returns tdl::Comment:
    'Note:' body=EString
;

Annotation returns tdl::Annotation:
    '@' key=[tdl::AnnotationType|Identifier]
    (':' value=EString)?
;

AnnotationType returns tdl::AnnotationType:
    AnnotationCommentFragment
    'AnnotationType' name=Identifier
    extension=Extension?
;

TestObjective returns tdl::TestObjective:
    AnnotationCommentFragment
    'TestObjective' name=Identifier
    (BEGIN
        ('Description:' description=EString)?
        ('References:' objectiveURI+=EString (',' objectiveURI+=EString)*)?
    END)?
;

Extension returns tdl::Extension:
    //NOTE: No annotations or comments
    'extends' extending=[tdl::PackageableElement|Identifier]
;

ConstraintType returns tdl::ConstraintType: 
    AnnotationCommentFragment
    'Constraint' name=Identifier
;

Constraint returns tdl::Constraint: 
    //NOTE: No annotations or comments
    type=[tdl::ConstraintType|Identifier]
    (':' quantifier+=LiteralValueUse (',' quantifier+=LiteralValueUse)+ )?
;

//# Data

//## Data Definition

DataResourceMapping returns tdl::DataResourceMapping:
    AnnotationCommentFragment
    'Use' resourceURI=EString
    'as' name=Identifier
;

DataElementMapping returns tdl::DataElementMapping:
    AnnotationCommentFragment
    'Map' mappableDataElement=[tdl::MappableDataElement|Identifier]
    ('to' elementURI=EString)?
    'in' dataResourceMapping=[tdl::DataResourceMapping|Identifier]
    'as' name=Identifier
    (BEGIN
        //TODO: separator 
        (parameterMapping+=ParameterMapping)*
    END)?
;

ParameterMapping returns tdl::ParameterMapping:
    AnnotationCommentFragment
    parameter=[tdl::Parameter|Identifier]
    '->' parameterURI=EString
;


SimpleDataType_Impl returns tdl::SimpleDataType:
    AnnotationCommentFragment
    ConstraintFragment?
    'Type' name=Identifier
;

SimpleDataInstance returns tdl::SimpleDataInstance:
    AnnotationCommentFragment
    dataType=[tdl::DataType|Identifier]
    name=Identifier
;

//TODO: alternative specification dialects?
StructuredDataType returns tdl::StructuredDataType:
    AnnotationCommentFragment
    ConstraintFragment?
    'Type' name=Identifier
    extension=Extension?
    '(' (member+=Member (',' member+=Member)*)? ')'
;

Member returns tdl::Member:
    AnnotationCommentFragment
    MemberConstraintFragment?
    (isOptional?='optional')?
    dataType=[tdl::DataType|Identifier]
    name=Identifier
;

//TODO: alternative specification dialects?
StructuredDataInstance returns tdl::StructuredDataInstance:
    AnnotationCommentFragment
    dataType=[tdl::DataType|Identifier]
    name=Identifier
    //TODO: use keywords for unassigned member treatment?
    (unassignedMember=UnassignedMemberTreatment)?
    '(' (memberAssignment+=MemberAssignment (',' memberAssignment+=MemberAssignment)*)? ')'
;

MemberAssignment returns tdl::MemberAssignment:
    AnnotationCommentFragment
    member=[tdl::Member|Identifier]
    '=' memberSpec=StaticDataUse
;

CollectionDataType returns tdl::CollectionDataType:
    AnnotationCommentFragment
    ConstraintFragment?
    'Collection' name=Identifier
    'of' itemType=[tdl::DataType|Identifier]
;

CollectionDataInstance returns tdl::CollectionDataInstance:
    AnnotationCommentFragment
    dataType=[tdl::DataType|Identifier] 
    name=Identifier
    //TODO: separators?
    '[' item+=StaticDataUse* ']'
;

ProcedureSignature returns tdl::ProcedureSignature:
    AnnotationCommentFragment
    'Signature' name=Identifier
    '(' parameter+=ProcedureParameter ( "," parameter+=ProcedureParameter)* ')'
;

ProcedureParameter returns tdl::ProcedureParameter:
    AnnotationFragment
    kind=ParameterKind
    dataType=[tdl::DataType|Identifier]
    name=Identifier
    WithCommentFragment?
;

enum ParameterKind returns tdl::ParameterKind:
    In = 'in' | Out = 'out' | Exception = 'exception';

//TODO: remove?
Parameter returns tdl::Parameter:
    Member | FormalParameter | ProcedureParameter;

FormalParameter returns tdl::FormalParameter:
    AnnotationFragment
    dataType=[tdl::DataType|Identifier]
    name=Identifier
    WithCommentFragment?
;

Variable returns tdl::Variable:
    AnnotationCommentFragment
    'variable' dataType=[tdl::DataType|Identifier]
    name=Identifier
    WithCommentFragment?
;

//TODO: remove?
Action returns tdl::Action:
    Action_Impl | Function;

Action_Impl returns tdl::Action:
    AnnotationCommentFragment
    'Action' name=Identifier
    ('(' formalParameter+=FormalParameter ( "," formalParameter+=FormalParameter)* ')')?
    (':' body=EString)?
;
    
Function returns tdl::Function:
    AnnotationCommentFragment
    'Function' name=Identifier
    ('(' formalParameter+=FormalParameter ( "," formalParameter+=FormalParameter)* ')')?
    'returns' returnType=[tdl::DataType|Identifier]
    (':' body=EString)?
;

enum UnassignedMemberTreatment returns tdl::UnassignedMemberTreatment:
    AnyValue = '?' | AnyValueOrOmit = '*';

PredefinedFunction returns tdl::PredefinedFunction:
    AnnotationCommentFragment
    'Predefined' 
    (name=PredefinedIdentifierBinary
        | name=PredefinedIdentifierNot
        | name=PredefinedIdentifierSize
    )
    ('returns' returnType=[tdl::DataType|Identifier])?
;

PredefinedIdentifierBinary returns ecore::EString:
    '+' | '-' | '*' | '/' | 'mod' 
  | '>' | '<' | '>=' | '<=' //TODO: fix extra '>=' also in TDLan2
  | '==' | '!=' | 'and' | 'or' | 'xor'
;

PredefinedIdentifierNot returns ecore::EString:
    'not'
;

PredefinedIdentifierSize returns ecore::EString:
    'size'
;

EnumDataType returns tdl::EnumDataType: 
    AnnotationCommentFragment
    'Enumerated' name=Identifier
    BEGIN
        //TODO: without type specification? separator?
        (value+=SimpleDataInstance)+
    END
;

//## Data Use

DataUse returns tdl::DataUse:
    //NOTE: No name, annotations or comments
    StaticDataUse | =>DynamicDataUse
;

ParameterBinding returns tdl::ParameterBinding:
    parameter=[tdl::Parameter|Identifier]
    '=' dataUse=DataUse
;

MemberReference returns tdl::MemberReference:
    member=[tdl::Member|Identifier] 
    (->'[' collectionIndex=DataUse ']')? 
;

CollectionReference returns tdl::MemberReference:
    '[' collectionIndex=DataUse ']'
;

StaticDataUse returns tdl::StaticDataUse:
    DataInstanceUse
    | AnonymousDataInstanceUse 
    | SpecialValueUse
    | LiteralValueUse
;

DataInstanceUse returns tdl::DataInstanceUse:
    (dataInstance=[tdl::DataInstance|Identifier]
    |
    ('a' dataType=[tdl::StructuredDataType|Identifier]))
    UnassignedFragment?
    ParameterBindingFragment?
    ReductionFragment
;

AnonymousDataInstanceUse returns tdl::DataInstanceUse:
    UnassignedFragment?
    ParameterBindingFragment
;

SpecialValueUse returns tdl::SpecialValueUse:
    OmitValue | AnyValue | AnyValueOrOmit 
;

AnyValue returns tdl::AnyValue:
    name='?'
    ('of' 'type' dataType=[tdl::DataType|Identifier])?
;

AnyValueOrOmit returns tdl::AnyValueOrOmit:
    name='*'
;

OmitValue returns tdl::OmitValue:
    name='omit'
;

LiteralValueUse returns tdl::LiteralValueUse:
    (value=STRING | intValue=BIGINTEGER | boolValue=BOOLEAN)
    //TODO: add option to have the type after wards without any keyword e.g. 100 s?
    ('of' 'type' dataType=[tdl::DataType|Identifier])?
;

DynamicDataUse returns tdl::DynamicDataUse:
    FunctionCall 
    | FormalParameterUse
    | VariableUse 
    | PredefinedFunctionCall
    | TimeLabelUse 
;

FunctionCall returns tdl::FunctionCall:
    //TODO: refine syntax
    'instance' 'returned' 'from' function=[tdl::Function|Identifier] 
    '(' BareParameterBindingFragment? ')'
    ReductionFragment
;

FormalParameterUse returns tdl::FormalParameterUse:
    'parameter' parameter=[tdl::FormalParameter|Identifier]
    ParameterBindingFragment?
    ReductionFragment
;

VariableUse returns tdl::VariableUse:
    //TODO: try to adjust syntax so that -> is not necessary
    componentInstance=[tdl::ComponentInstance|Identifier]
    '->' variable=[tdl::Variable|Identifier]
    ParameterBindingFragment?
    ReductionFragment
;

PredefinedFunctionCall returns tdl::PredefinedFunctionCall:
    PredefinedFunctionCallSize
    | PredefinedFunctionCallNot 
    | PredefinedFunctionCallBinary
;

PredefinedFunctionCallSize returns tdl::PredefinedFunctionCall:
    function=[tdl::PredefinedFunction|PredefinedIdentifierSize] 
    '(' actualParameters+=DataUse ')'
;


PredefinedFunctionCallNot returns tdl::PredefinedFunctionCall:
    function=[tdl::PredefinedFunction|PredefinedIdentifierNot] 
    '(' actualParameters+=DataUse ')'
;

PredefinedFunctionCallBinary returns tdl::PredefinedFunctionCall:
    '('
    actualParameters+=DataUse 
    function=[tdl::PredefinedFunction|PredefinedIdentifierBinary] 
    actualParameters+=DataUse
    ')'
;

//# Time

Time returns tdl::Time:
    AnnotationCommentFragment
    'Time' name=Identifier
;

TimeLabel returns tdl::TimeLabel:
    //NOTE: No annotations or comments
    name=Identifier
;

TimeLabelUse returns tdl::TimeLabelUse:
    //NOTE: No reduction and parameter binding (constraint)
    //TODO: refine syntax
    'time' 'label' timeLabel=[tdl::TimeLabel|Identifier]
    ('.' kind=TimeLabelUseKind)?
;


enum TimeLabelUseKind returns tdl::TimeLabelUseKind:
    Last = 'last' | Previous = 'previous' | First = 'first'
;

TimeConstraint returns tdl::TimeConstraint:
    //NOTE: No annotations or comments
    timeConstraintExpression=DataUse
;

Timer returns tdl::Timer:
    AnnotationCommentFragment
    'timer' name=Identifier
;

TimeOperation returns tdl::TimeOperation: 
    Wait | Quiescence
;

Wait returns tdl::Wait: 
    AtomicPrefixFragment
    'wait' 'for' period=DataUse
    ( 'on' componentInstance=[tdl::ComponentInstance|Identifier])
;

Quiescence returns tdl::Quiescence: 
    AtomicPrefixFragment
    'quiet' 'for' period=DataUse 
    (  'on' (
            componentInstance=[tdl::ComponentInstance|Identifier] 
            | gateReference=[tdl::GateReference|IdentifierDot]
        )
    )?
;


TimerOperation returns tdl::TimerOperation: 
    TimerStart | TimerStop | TimeOut
;

TimerStart returns tdl::TimerStart:
    AtomicPrefixFragment
    'start' componentInstance=[tdl::ComponentInstance|Identifier]
    //TODO: use '->' instead?
    '.' timer=[tdl::Timer|Identifier]
    'for' period=DataUse
;

TimerStop returns tdl::TimerStop:
    AtomicPrefixFragment
    'stop' componentInstance=[tdl::ComponentInstance|Identifier]
    '.' timer=[tdl::Timer|Identifier]
;

TimeOut returns tdl::TimeOut:
    AtomicPrefixFragment
    'time' 'out' 'on' componentInstance=[tdl::ComponentInstance|Identifier]
    '.' timer=[tdl::Timer|Identifier]
;

//# Test Configuration

GateType returns tdl::GateType:
    AnnotationCommentFragment
    kind=GateTypeKind
    'Gate' name=Identifier
    (extension=Extension)?
    'accepts' dataType+=[tdl::DataType|Identifier] ( "," dataType+=[tdl::DataType|Identifier])*
;

enum GateTypeKind returns tdl::GateTypeKind:
    Message = 'Message' | Procedure = 'Procedure';

GateInstance returns tdl::GateInstance:
    AnnotationCommentFragment
    'gate' type=[tdl::GateType|Identifier]
    name=Identifier
    WithCommentFragment?  
;

ComponentType returns tdl::ComponentType:
    AnnotationCommentFragment
    'Component' name=Identifier
    (extension=Extension)?
    BEGIN
        (timer+=Timer)*
        (variable+=Variable)*
        (gateInstance+=GateInstance)* 
    END
;

ComponentInstance returns tdl::ComponentInstance:
    AnnotationCommentFragment
    'create' role=ComponentInstanceRole
    name=Identifier
    'of' 'type' type=[tdl::ComponentType|Identifier]
;

enum ComponentInstanceRole returns tdl::ComponentInstanceRole:
    SUT = 'SUT' | Tester = 'Tester';

GateReference returns tdl::GateReference:
    //NOTE: No annotations or comments
    (name=Identifier '=')? //TODO: make sure this is handled properly in the scoping
    component=[tdl::ComponentInstance|Identifier]
    '.' 
    gate=[tdl::GateInstance|Identifier]
;

Connection returns tdl::Connection:
    AnnotationCommentFragment
    'connect' endPoint+=GateReference 
    'to' endPoint+=GateReference
    WithNameFragment?
;


TestConfiguration returns tdl::TestConfiguration:
    AnnotationCommentFragment
    'Configuration' name=Identifier
    BEGIN
        (componentInstance+=ComponentInstance)* 
        (connection+=Connection)* 
    END
;

//# Test Behaviour

//## Test Description

TestDescription returns tdl::TestDescription:
    TDPrefixFragment
    //TODO: consider better way to specify / indicate local ordering
    ('TestDescription' | isLocallyOrdered?='Test') 
    name=Identifier
    ('(' formalParameter+=FormalParameter ( "," formalParameter+=FormalParameter)* ')' )?
    'uses' testConfiguration=[tdl::TestConfiguration|Identifier]
    (behaviourDescription=BehaviourDescription)?
;

BehaviourDescription returns tdl::BehaviourDescription:
    //NOTE: No annotations or comments
    behaviour=Behaviour
;

//## Combined Behaviour

Behaviour returns tdl::Behaviour:
    CombinedBehaviour | AtomicBehaviour
    //NOTE: Excluded via constraint:
    //PeriodicBehaviour
    //DefaultBehaviour | InterruptBehaviour | 
;

Block returns tdl::Block:
    //NOTE: No annotations or comments
    ('[' guard+=LocalExpression ( "," guard+=LocalExpression)* ']' )?
    BEGIN
        behaviour+=Behaviour+ 
    END
;

LocalExpression returns tdl::LocalExpression:
    //NOTE: No annotations or comments
    expression=DataUse
    ('on' scope=[tdl::ComponentInstance|Identifier])?
;

LocalLoopExpression returns tdl::LocalExpression:
    //NOTE: No annotations or comments
    expression=DataUse 'times'
    ('on' scope=[tdl::ComponentInstance|Identifier])?
;


CombinedBehaviour returns tdl::CombinedBehaviour:
    (SingleCombinedBehaviour | MultipleCombinedBehaviour)
    =>WithCombinedFragment?
;

SingleCombinedBehaviour returns tdl::SingleCombinedBehaviour:
    CompoundBehaviour
    | BoundedLoopBehaviour
    | UnboundedLoopBehaviour 
    | OptionalBehaviour
;

CompoundBehaviour returns tdl::CompoundBehaviour:
    AnnotationFragment
    block=Block
;

BoundedLoopBehaviour returns tdl::BoundedLoopBehaviour:
    AnnotationFragment
    'repeat' numIteration+=LocalLoopExpression (',' numIteration+=LocalLoopExpression)*
    block=Block
;

UnboundedLoopBehaviour returns tdl::UnboundedLoopBehaviour:
    AnnotationFragment
    'repeat' block=Block
;

OptionalBehaviour returns tdl::OptionalBehaviour:
    AnnotationFragment
    'optionally' block=Block
;

MultipleCombinedBehaviour returns tdl::MultipleCombinedBehaviour:
    ConditionalBehaviour 
    | AlternativeBehaviour
    | ParallelBehaviour
;

ConditionalBehaviour returns tdl::ConditionalBehaviour:
    AnnotationFragment
    'if' block+=Block
    (=>('else' block+=Block)
    | (('else' 'if' block+=Block)* 
       ('else' block+=Block)))? 
;

AlternativeBehaviour returns tdl::AlternativeBehaviour:
    AnnotationFragment
    'alternatively' block+=Block
    //TODO: enforce 2+? 
    ('or' block+=Block)* 
;

ParallelBehaviour returns tdl::ParallelBehaviour:
    AnnotationFragment
    'run' block+=Block 
    //TODO: enforce 2+?
    //TODO: simplify keywords / syntax
    ('in' 'parallel' 'to' block+=Block)?
    ('and' block+=Block)*
;

ExceptionalBehaviour returns tdl::ExceptionalBehaviour:
    DefaultBehaviour | InterruptBehaviour;

DefaultBehaviour returns tdl::DefaultBehaviour:
    AnnotationFragment
    'default' 
    ('on' guardedComponent=[tdl::ComponentInstance|Identifier])?
    block=Block
    WithBehaviourFragment?
;

InterruptBehaviour returns tdl::InterruptBehaviour:
    AnnotationFragment
    'interrupt'
    ('on' guardedComponent=[tdl::ComponentInstance|Identifier])?
    block=Block
    WithBehaviourFragment?
;

PeriodicBehaviour returns tdl::PeriodicBehaviour:
    AnnotationFragment
    'every'
    (period+=LocalExpression (',' period+=LocalExpression)*)
    block=Block
    WithBehaviourFragment?
;

//## Atomic Behaviour

AtomicBehaviour returns tdl::AtomicBehaviour:
    (TimerOperation
    | TimeOperation
    | Break | Stop
    | VerdictAssignment | Assertion
    | Interaction
    | TestDescriptionReference 
    | ActionBehaviour 
    | Assignment)
    WithAtomicFragment? 
;

Break returns tdl::Break:
    {tdl::Break}
    AtomicPrefixFragment
    'break'
;

Stop returns tdl::Stop:
    {tdl::Stop}
    AtomicPrefixFragment
    'terminate'
;

VerdictAssignment returns tdl::VerdictAssignment:
    AtomicPrefixFragment
    'set' 'verdict' 'to' verdict=DataUse
;

Assertion returns tdl::Assertion:
    {tdl::Assertion}
    AtomicPrefixFragment
    'assert' condition=DataUse
    ('on' componentInstance=[tdl::ComponentInstance|Identifier])?
    //TODO: simplify syntax
    ('otherwise' 'set' 'verdict' 'to' otherwise=DataUse)?
;

Interaction returns tdl::Interaction: 
    Message | ReceiveMessage | ProcedureCall | ProcedureCallResponse
;

Message returns tdl::Message:
    AtomicPrefixFragment
    sourceGate=[tdl::GateReference|IdentifierDot]
    ('sends' | (isTrigger?='triggers')) 
    argument=DataUse
    'to' target+=TargetMessage ( "," target+=TargetMessage)* 
;

TargetMessage returns tdl::Target:
    //NOTE: No name, annotations or comments
    targetGate=[tdl::GateReference|IdentifierDot]
    (valueAssignment+=ValueAssignmentMessage)?
;

ValueAssignmentMessage returns tdl::ValueAssignment:
    //NOTE: No name, annotations or comments
    //TODO: check what the correct syntax shall be
    //'where' 'it' 'is' 
    'assigned' 'to' variable=[tdl::Variable|Identifier]
;


ReceiveMessage returns tdl::Message:
    AtomicPrefixFragment
    //NOTE: Only single target possible
    target+=ReceiveTargetMessage //( "," target+=ReceiveTargetMessage)*
    'receives' (isTrigger?='trigger')? 
    argument=DataUse
    'from' sourceGate=[tdl::GateReference|IdentifierDot]
;

ReceiveTargetMessage returns tdl::Target:
    //NOTE: No name, annotations or comments
    (valueAssignment+=ReceiveValueAssignmentMessage)?
    targetGate=[tdl::GateReference|IdentifierDot]
;

ReceiveValueAssignmentMessage returns tdl::ValueAssignment:
    //NOTE: No name, annotations or comments
    variable=[tdl::Variable|Identifier] '='
;



ProcedureCall returns tdl::ProcedureCall:
    AtomicPrefixFragment
    sourceGate=[tdl::GateReference|IdentifierDot]
    'calls' signature=[tdl::ProcedureSignature|Identifier]
    '('argument+=ParameterBinding ("," argument+=ParameterBinding)*')'
    'on' target+=TargetProcedure
;

ProcedureCallResponse returns tdl::ProcedureCall:
    AtomicPrefixFragment
    //TODO: this shall probably be switched
    (replyTo=[tdl::ProcedureCall|Identifier] ':')?
    sourceGate=[tdl::GateReference|IdentifierDot]
    'responds' 'with' signature=[tdl::ProcedureSignature|Identifier]
    '('argument+=ParameterBinding ("," argument+=ParameterBinding)*')'
    'to' target+=TargetProcedure
;

TargetProcedure returns tdl::Target:
    //NOTE: No name, annotations or comments
    targetGate=[tdl::GateReference|IdentifierDot]
    //TODO: check what the correct syntax shall be
    (valueAssignment+=ValueAssignmentProcedure ("," valueAssignment+=ValueAssignmentProcedure)*)?
;

ValueAssignmentProcedure returns tdl::ValueAssignment:
    //NOTE: No name, annotations or comments
    //TODO: check what the correct syntax shall be
    'where' parameter=[tdl::Parameter|Identifier] 'is' 'assigned' 'to' variable=[tdl::Variable|Identifier]
;
    
TestDescriptionReference returns tdl::TestDescriptionReference:
    AtomicPrefixFragment
    'execute' testDescription=[tdl::TestDescription|Identifier]
    ('(' actualParameter+=ParameterBinding ( "," actualParameter+=ParameterBinding)* ')' )?
    (BEGIN
        componentInstanceBinding+=ComponentInstanceBinding (',' componentInstanceBinding+=ComponentInstanceBinding)+
    END)?
;

ComponentInstanceBinding returns tdl::ComponentInstanceBinding:
    AnnotationCommentFragment
    'bind' formalComponent=[tdl::ComponentInstance|Identifier]
    'to' actualComponent=[tdl::ComponentInstance|Identifier]
;

ActionBehaviour returns tdl::ActionBehaviour: 
    ActionReference | InlineAction
;

ActionReference returns tdl::ActionReference:
    AtomicPrefixFragment
    'perform' action=[tdl::Action|Identifier]
    ('(' actualParameter+=ParameterBinding ( "," actualParameter+=ParameterBinding)* ')' )?
    ('on' componentInstance=[tdl::ComponentInstance|Identifier])?
;

InlineAction returns tdl::InlineAction:
    AtomicPrefixFragment
    'perform' 'action' ':' body=EString
    ('on' componentInstance=[tdl::ComponentInstance|Identifier])?
;

Assignment returns tdl::Assignment:
    AtomicPrefixFragment
    variable=VariableUse
    '=' expression=DataUse
    ('on' componentInstance=[tdl::ComponentInstance|Identifier])?
;


//# Auxiliary

//## Fragments

//Element : Abstract
fragment AnnotationFragment returns tdl::Element:
    (annotation+=Annotation)*
;

fragment AnnotationCommentFragment returns tdl::Element:
    (comment+=Comment)*
    (annotation+=Annotation)*
;

fragment NameFragment returns tdl::Element:
    'Name:' name=Identifier
;

//Use: inline block - within other blocks, e.g. with other contents 
fragment WithCommentFragment returns tdl::Element:
    'with'
    BEGIN
        (comment+=Comment)+
        //TODO: Shall this be '*'? Ambiguity?
    END
;

fragment WithNameFragment returns tdl::Element:
    'with'
    BEGIN
        NameFragment
    END
;

//DataType: Abstract

fragment ConstraintFragment returns tdl::DataType:
    //TODO: put braces into constraint?
    LBrace (constraint+=Constraint)+ RBrace
;

//Member

fragment MemberConstraintFragment returns tdl::Member:
    //TODO: put braces into constraint?
    LBrace (constraint+=Constraint)+ RBrace
;


//DataUse: Abstract

fragment ReductionFragment returns tdl::DataUse:
    (->reduction+=CollectionReference)?
    ('.' reduction+=MemberReference)*
;

fragment ParameterBindingFragment returns tdl::DataUse:
    =>'(' BareParameterBindingFragment ')'
;

fragment BareParameterBindingFragment returns tdl::DataUse:
    =>(argument+=ParameterBinding ( "," argument+=ParameterBinding)*)
;

//DataInstanceUse: Abstract

fragment UnassignedFragment returns tdl::DataInstanceUse:
    //TODO: added angle brackets for disambiguation
    =>'<' unassignedMember=UnassignedMemberTreatment '>'
;

//Behaviour : Abstract

fragment WithBehaviourFragment returns tdl::Behaviour:
    'with'
    BEGIN
        NameFragment?
        ObjectiveFragment?
        (comment+=Comment)*
    END
;

fragment ObjectiveFragment returns tdl::Behaviour:
    'Objective:' testObjective+=[tdl::TestObjective|Identifier] ( "," testObjective+=[tdl::TestObjective|Identifier])*
;

//TestDescription

fragment TDPrefixFragment returns tdl::TestDescription:
    (comment+=Comment)*
    TDObjectiveFragment?
    (annotation+=Annotation)*
;

fragment TDObjectiveFragment returns tdl::TestDescription:
    'Objective:' testObjective+=[tdl::TestObjective|Identifier] ( "," testObjective+=[tdl::TestObjective|Identifier])*
;

//AtomicBehaviour : Abstract

fragment AtomicPrefixFragment returns tdl::AtomicBehaviour:
    ObjectiveFragment?
    (comment+=Comment)*
    (annotation+=Annotation)*
;

fragment WithAtomicFragment returns tdl::AtomicBehaviour:
    'with'
    BEGIN
        NameFragment?
        TimeLabelFragment?
        TimeConstraintFragment?
    END
;

fragment TimeLabelFragment returns tdl::AtomicBehaviour:
    timeLabel=TimeLabel '=now' 
;

fragment TimeConstraintFragment returns tdl::AtomicBehaviour:
    LBrace timeConstraint+=TimeConstraint ( "," timeConstraint+=TimeConstraint)* RBrace
;


//CombinedBehaviour : Abstract

fragment WithCombinedFragment returns tdl::CombinedBehaviour:
    'with'
    BEGIN
        NameFragment?
        ObjectiveFragment?
        (comment+=Comment)*
        //TODO: periodic / exceptional as separate fragment(s) for combined behaviour?
        (periodic+=PeriodicBehaviour)*
        (exceptional+=ExceptionalBehaviour)*
    END
;


//## ValueConverter Rules (Pseudo-terminals)

//TODO: why does this strip quotes when it is a single word?
EString returns ecore::EString:
    STRING
    // | ID
;

LiteralValue returns ecore::EString:
    STRING
    | ('-'? INT ('.' INT)?)
;


Identifier returns ecore::EString:
    ID
;

IdentifierDot returns ecore::EString:
    ID '.' ID
;

QIdentifier returns ecore::EString:
    ID ('.' ID)*
;

//Retain Braces even in indentation-based
LBrace:
    BEGIN
;

RBrace:
    END
;

BIGINTEGER returns ecore::EBigInteger:
	INT
;

//## Terminals

terminal BOOLEAN returns ecore::EBoolean:
	'true' | 'false'
;

[bookmark: _Toc83250438]Annex B (informative):
Examples	Comment by Philip Makedonski: Include in overview. Complete examples (from Annex B of part 1 and others)	Comment by Philip Makedonski: Also electronic?

[bookmark: _Toc83250439]B.0	Overview
This annex provides several examples to illustrate how the different elements of the TDL Textual Syntax can be used and demonstrates the applicability of TDL in several different areas.
The first example in clause B.1 demonstrates the usage of data-related concepts.
The second example in clause B.2 shows a scenario when a 'Tester' performs a test scenario on one interface of the 'SUT'. The example is taken from ETSI TS 136 523-1 [i.1].
The third example in clause B.3 provides an example for interoperability testing in IMS. The example is taken from ETSI TS 186 011-2 [i.2].

[bookmark: _Toc83250440]B.1	…

[bookmark: _Toc83250441]B.2	…


[bookmark: _Toc83250442]History
	Document history

	V1.1.1
	September 2021
	Early draft



ETSI
image1.jpeg




